

```
[> # Maple sample file (text mode) for J. ISCIE by M. Kanno 2011.03
> # システム/制御/情報 2011年5月号 (Vol. 55 No. 5)
> # 数式処理を用いたパラメータを有する線形制御系の解析 管野 政明
> # Section 2 - 基本操作と伝達関数
```

```
> restart;
```

```
> ## 基本操作
```

```
> f := x^10 - 1;
```

$$f := x^{10} - 1 \quad (1)$$

```
> g := factor(f);
```

$$g := (x - 1) (x + 1) (x^4 + x^3 + x^2 + x + 1) (x^4 - x^3 + x^2 - x + 1) \quad (2)$$

```
> h := (x+1)^2 - x^2 + 1; simplify(h);
```

$$h := (x + 1)^2 - x^2 + 1 \\ 2x + 2 \quad (3)$$

```
> g; expand(g);
```

$$(x - 1) (x + 1) (x^4 + x^3 + x^2 + x + 1) (x^4 - x^3 + x^2 - x + 1) \\ x^{10} - 1 \quad (4)$$

```
> (x^2 - 1) / (x - 1); normal(%);
```

$$\frac{x^2 - 1}{x - 1} \\ x + 1 \quad (5)$$

```
> ## 伝達関数
```

```
> P := 2 / (s + 2); K := 1 / s;
```

$$P := \frac{2}{s + 2} \\ K := \frac{1}{s} \quad (6)$$

```
> Tcl := P*K / (1 + P*K); Tcl := simplify(Tcl);
```


$$Tcl := \frac{2}{(s + 2) s \left( 1 + \frac{2}{(s + 2) s} \right)} \\ Tcl := \frac{2}{s^2 + 2s + 2} \quad (7)$$

```
> with(inttrans):
```

```
> Tcl_t := invlaplace(Tcl*1/s, s, t);
```

$$Tcl\_t := 1 - e^{-t} (\cos(t) + \sin(t)) \quad (8)$$

```
> plot(Tcl_t, t = 0..10);
```



```
> limit(Tc1_t, t = infinity);
```

$$1$$
(9)

```
> diff(Tc1_t, t); Tc12_t := simplify(%);
```

$$e^{-t}(\cos(t) + \sin(t)) - e^{-t}(-\sin(t) + \cos(t))$$

$$Tc12_t := 2 e^{-t} \sin(t)$$
(10)

```
> int(Tc12_t^2, t = 0..infinity);
```

$$\frac{1}{2}$$
(11)

```
> g_t := invlaplace(1 / (s + a), s, t);
```

$$g_t := e^{-at}$$
(12)

```
> limit(g_t, t = infinity);
```

$$\lim_{t \rightarrow \infty} e^{-at}$$
(13)

```
> limit(g_t, t = infinity) assuming a > 0;
```

$$0$$
(14)

```
> invlaplace(1 / (s^3 + 2*s^2 + 3*s + 4), s, t);
```

$$\frac{1}{20} \sum_{\alpha = \text{RootOf}(\_Z^3 + 2\_Z^2 + 3\_Z + 4)} e^{-\alpha t} (-\_ \alpha + \_ \alpha^2)$$
(15)

```
> evalf(%);
```

$$0.2187602960 e^{-1.650629191 t} + (-0.1093801480 + 0.1043649860 I) e^{(-0.1746854043 - 1.546868887 I) t} + (-0.1093801480$$
(16)

$$-0.1043649860 I) e^{(-0.1746854043 + 1.546868887 I) t}$$

```
> # Section 3 - パラメータを有する線形系の安定性
> with(PolynomialTools): with(LinearAlgebra):
> read "HurwitzMatrix.maple";
> [seq(a||i, i = 0..5)]; p := sort(FromCoefficientList(%, s));
```

$$p := a5 s^5 + a4 s^4 + a3 s^3 + a2 s^2 + a1 s + a0 \quad (17)$$

```
> Hn := HurwitzMatrix(p, s);
```

$$Hn := \begin{bmatrix} a4 & a2 & a0 & 0 & 0 \\ a5 & a3 & a1 & 0 & 0 \\ 0 & a4 & a2 & a0 & 0 \\ 0 & a5 & a3 & a1 & 0 \\ 0 & 0 & a4 & a2 & a0 \end{bmatrix} \quad (18)$$

```
> # if 文を用いたプロシージャ
> read "HurwitzMatrix2.maple";
> Hn2 := HurwitzMatrix2(p, s);
```

$$Hn2 := \begin{bmatrix} a4 & a2 & a0 & 0 & 0 \\ a5 & a3 & a1 & 0 & 0 \\ 0 & a4 & a2 & a0 & 0 \\ 0 & a5 & a3 & a1 & 0 \\ 0 & 0 & a4 & a2 & a0 \end{bmatrix} \quad (19)$$

```
> Hn - Hn2;
```

$$\begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix} \quad (20)$$

```
> # 同じ結果が得られた
```

```
> p := convert(s^3 + (1.03 + q)*s^2 + (1.02 + 3*q)*s + (1.01 + 7*q),
rational, exact);
```

$$p := s^3 + \left(\frac{103}{100} + q\right) s^2 + \left(\frac{51}{50} + 3q\right) s + \frac{101}{100} + 7q \quad (21)$$

```
> Hn := HurwitzMatrix(p, s);
```

$$Hn := \begin{bmatrix} \frac{103}{100} + q & \frac{101}{100} + 7q & 0 \\ 1 & \frac{51}{50} + 3q & 0 \\ 0 & \frac{103}{100} + q & \frac{101}{100} + 7q \end{bmatrix} \quad (22)$$

```
> f := sort(Determinant(Hn));
```

$$f := 21 q^3 - \frac{86}{5} q^2 - \frac{26347}{10000} q + \frac{20503}{500000} \quad (23)$$

```
> roots_f := sort(realroot(f, 10^(-8))); evalf(roots_f);
roots_f := [[[-19365701, -4841425], [1913879, 239235], [127382531,
31845633], [134217728, -33554432], [134217728, 16777216], [134217728,
33554432]]]
[[-0.1442857161, -0.1442857087], [0.01425950974, 0.01425951719], [0.9490738139,
0.9490738213]] (24)
```

```
> subs(q = roots_f[1][1] - 1, p); Hurwitz(%, s);
s^3 - 383479229/3355443200 s^2 - 8096205111/3355443200 s - 23488102443/3355443200
false (25)
```

```
> Hurwitz(subs(q = (roots_f[1][2] + roots_f[2][1])/2, p), s);
true (26)
```

```
> Hurwitz(subs(q = (roots_f[2][2] + roots_f[3][1])/2, p), s);
false (27)
```

```
> Hurwitz(subs(q = roots_f[3][2] + 1, p), s);
true (28)
```

```
> # Section 4 - パラメータを有する線形系のH2ノルム
```

```
> A := Matrix([[-a, -1], [1, 0]]); B := Vector([1, 0]); C := Vector(
[2, -1], orientation = row);
```

$$A := \begin{bmatrix} -a & -1 \\ 1 & 0 \end{bmatrix}$$

$$B := \begin{bmatrix} 1 \\ 0 \end{bmatrix}$$

$$C := \begin{bmatrix} 2 & -1 \end{bmatrix}$$

(29)

```
> sort(simplify(C.(s*IdentityMatrix(2) - A)^(-1).B), s);
```

$$\frac{2s-1}{s^2+as+1}$$

(30)

```
> Lo := Matrix(2, symbol = l, shape = symmetric);
```

$$Lo := \begin{bmatrix} l_{1,1} & l_{1,2} \\ l_{1,2} & l_{2,2} \end{bmatrix}$$

(31)

```
> ## solve の使い方
```

```
> solve(x^2 - 2);
```

$$\sqrt{2}, -\sqrt{2}$$

(32)

```
> solve(x^2 - a, x);
```

$$\sqrt{a}, -\sqrt{a}$$

(33)

```
> solve(x^2 - a);
```

$$\{a=x^2, x=x\}$$

(34)

```
> solve(x^2 - a, {x});
```

$$\{x=\sqrt{a}\}, \{x=-\sqrt{a}\}$$

(35)

```
> solve({x - y = 0, 2*x + y = 3});
```

$$\{x=1, y=1\}$$

(36)

```
> ## solve の使い方 - 終わり
```

```
> Lyap := Transpose(A).Lo + Lo.A + Transpose(C).C;
```

$$Lyap := \begin{bmatrix} -2 a l_{1,1} + 2 l_{1,2} + 4 & -a l_{1,2} + l_{2,2} - l_{1,1} - 2 \\ -a l_{1,2} + l_{2,2} - l_{1,1} - 2 & -2 l_{1,2} + 1 \end{bmatrix} \quad (37)$$

```
> convert(Lo, set);
```

$$\{l_{1,1}, l_{1,2}, l_{2,2}\} \quad (38)$$

```
> sol := solve(convert(Lyap, set), convert(Lo, set));
```

$$sol := \left\{ l_{1,1} = \frac{5}{2a}, l_{1,2} = \frac{1}{2}, l_{2,2} = \frac{1}{2} \frac{a^2 + 5 + 4a}{a} \right\} \quad (39)$$

```
> subs(sol, Lo); Transpose(B).%B;
```

$$\begin{bmatrix} \frac{5}{2a} & \frac{1}{2} \\ \frac{1}{2} & \frac{1}{2} \frac{a^2 + 5 + 4a}{a} \end{bmatrix}$$

$$\frac{5}{2a}$$

(40)

```
> # end
```